

A factory produces three types of drinking glasses in batches of 1000. The batches may be incomplete, and we assume no setup cost.

Deliveries to distributors have been scheduled for the next 12 weeks:

Week	1	2	3	4	5	6	7	8	9	10	11	12
1	20	22	18	35	17	19	23	20	29	30	28	32
2	17	19	23	20	11	10	12	34	21	23	30	12
3	18	35	17	10	9	21	23	15	10	0	13	17

Production of a batch uses three scarce resources:

- **labor** (180 man-hours/week available)
- **machine time**, (200 machine hours/week available), and
- **storage space** (400 m³ available).

The resource requirements per batch are shown in the table, as well as the cost to produce and store a batch for one week.

Type	Prod'n Cost (\$) C	Storage Cost (\$) S	Initial Stock	Final Stock	man-hours L	machine-hours M	Storage (m ³) W
1	100	25	50	10	3	2	4
2	80	28	20	10	3	1	5
3	110	25	0	10	3	4	5

Some of the products are already in stock, and at the end of the 12 weeks, the factory manager wants to have 10 batches of each product in stock.

Schedule the factory production for the next 12 weeks.

Define variables

P_{it} = number of batches of product i ($i=1,2,3$) to be produced in week t ($t=1,2,\dots,12$).

I_{it} = number of batches of product i ($i=1,2,3$) in storage at beginning of week t ($t=1,2,\dots,13$).

Minimize $\sum_{i=1}^3 \sum_{j=1}^{12} C_i P_{ij} + \sum_{i=1}^3 \sum_{j=1}^{12} S_i I_{ij}$ Costs consist of production and storage costs

subject to

$$I_{11} = 50, \quad I_{21} = 20, \quad I_{31} = 0 \quad \text{Initial inventory levels}$$

$$I_{1,13} = I_{2,13} = I_{3,13} = 10 \quad \text{Final inventory levels}$$

for each week $j=1,2,\dots,12$:

$$\sum_{i=1}^3 L_i P_{ij} \leq 180 \quad \text{Labor availability constraint}$$

$$\sum_{i=1}^3 M_i P_{ij} \leq 200 \quad \text{Machine availability constraint}$$

$$\sum_{i=1}^3 W_i I_{ij} \leq 400 \quad \text{Warehouse space constraint}$$

Material balance constraints:

for each week $j=1,2,\dots,12$:

$$\begin{aligned} P_{ij} + I_{ij} &= D_{ij} + I_{i,j+1} \\ \text{production} + \text{inventory} &= \text{demand} + \text{next week's inventory} \end{aligned}$$

Lingo model (with sets):

```

MODEL: ! Drinking glass production scheduling;

SETS:
  TYPE/1..3/: C, S, L, M, W, INITIAL, FINAL;
  WEEK/1..13/;
  SCHEDULE (TYPE,WEEK): P, INV, DEMAND;
ENDSETS

DATA: ! week: 1 2 3 4 5 6 7 8 9 10 11 12 13;
  DEMAND =
    20 22 18 35 17 19 23 20 29 30 28 32 0
    17 19 23 20 11 10 12 34 21 23 30 12 0
    18 35 17 10 9 21 23 15 10 0 13 17 0;

  C = 100 80 110;           ! production costs per batch;
  S = 25 28 25;           ! storage costs per batch;
  L = 3 3 3;              ! labor requirement per batch;
  M = 2 1 4;              ! machine hours per batch;
  W = 4 5 5;              ! warehouse space/batch;
  INITIAL= 50 20 0;       ! initial inventory levels;
  FINAL= 10 10 10;       ! final inventory levels;
  MANHRS = 180;           ! labor available per week;
  MACHINEHRS = 200;      ! machine hours per week;
  SPACE = 400;            ! warehouse space available;
ENDDATA

```

```

! Objective: minimize total cost;
MIN = @SUM(SCHEDULE(I,J)|J #LE# 12: C(I)*P(I,J) ) ! production cost ;
      + @SUM(SCHEDULE(I,J): S(I)*INV(I,J)); ! storage cost ;

! Material balance equations ;
@FOR(TYPE(I):
  @FOR(WEEK(J)|J #LE# 12:
    P(I,J) + INV(I,J) = DEMAND(I,J) + INV(I,J+1);
  );
);

! Resource usage limitations;
@FOR(WEEK(J):
  @SUM(TYPE(I): L(I)*P(I,J) ) <= MANHRS;
  @SUM(TYPE(I): M(I)*P(I,J) ) <= MACHINEHRS;
  @SUM(TYPE(I): W(I)*INV(I,J) ) <= SPACE;
);

! boundary conditions;
@FOR(TYPE(I):
  INV(I,1) = INITIAL(I);
  INV(I,13) = FINAL(I);
);

END

```

Variable	Value
P(1, 2)	4.000000
P(1, 3)	10.000000
P(1, 4)	40.000000
P(1, 5)	40.000000
P(1, 6)	29.000000
P(1, 7)	2.000000
P(1, 8)	26.000000
P(1, 9)	2.000000
P(1, 10)	32.000000
P(1, 11)	30.000000
P(1, 12)	38.000000
P(2, 2)	16.000000
P(2, 3)	23.000000
P(2, 4)	20.000000
P(2, 5)	11.000000
P(2, 6)	10.000000
P(2, 7)	12.000000
P(2, 8)	34.000000
P(2, 9)	21.000000
P(2, 10)	23.000000
P(2, 11)	30.000000
P(2, 12)	22.000000

Variable	Value
P(3, 1)	18.000000
P(3, 2)	35.000000
P(3, 3)	27.000000
P(3, 5)	9.000000
P(3, 6)	21.000000
P(3, 7)	46.000000
P(3, 9)	37.000000
P(3, 10)	5.000000

Batch Production Schedule

Variable	Value
INV(1, 1)	50.000000
INV(1, 2)	30.000000
INV(1, 3)	12.000000
INV(1, 4)	4.000000
INV(1, 5)	9.000000
INV(1, 6)	32.000000
INV(1, 7)	42.000000
INV(1, 8)	21.000000
INV(1, 9)	27.000000
INV(1, 11)	2.000000
INV(1, 12)	4.000000
INV(1, 13)	10.000000
INV(2, 1)	20.000000
INV(2, 2)	3.000000
INV(2, 13)	10.000000
INV(3, 4)	10.000000
INV(3, 8)	23.000000
INV(3, 9)	8.000000
INV(3, 10)	35.000000
INV(3, 11)	40.000000
INV(3, 12)	27.000000
INV(3, 13)	10.000000

Numbers of batches in storage